ECE 2 - Année 2021-2022
ENC Bessières
Mathématiques - F. Gaunard
http://frederic.gaunard.com

Devoir Maison n°4bis

Éléments de solution

Exercice 1

(1) Pour chaque terme de la liste entre le premier et l'avant-dernier, on regarde tous les termes qui suivent. Si l'un d'entre eux est strictement plus petit, on les permute.

```
function T=tri(L)
    N=length(L)
    for i=1:N-1
        x=i
        for j=i+1:N
            if L(j)<L(x) then
                x=j
            end
        end
        aux=L(x)
        L(x)=L(i)
        L(i)=aux
    end
    T=L
endfunction
```

(2) Le principe de concaténation des listes qui apparait dans le programme précédent permet sans difficulté d'écrire celui-demandé. Pour information, ce programme était notamment demandé dans le sujet **ESSEC II 2020** et dans le dernier devoir surveillé (DS n°3).

```
function [y,L]=selection(U)
    N=length(U)
    j=grand(1,1,'uin', 1, N)
    y=U(j)
    L=[U(1:j-1), U(j+1: N)]
 endfunction
```

(3) On complète:

```
function x=X(n)
    U=1:n
    L=zeros(1,3)
    for i=1:3
        [k,U]=selection(U)
        L(i)=k
    end
    L=tri(L)
    x=L(2)
endfunction
```

(4) On modélise la situation par un tirage simultané de trois boules. Il y a $\binom{5}{3}$ façons de choisir ces trois boules. Si $[X=k]$, cela veut dire qu'une des deux autres boules a une valeur strictement inférieure à k et la dernière une valeur strictement supérieure. Il y a $k-1$ façons de choisir la valeur de la boule plus petite et $5-k$ façons de choisir la boule plus grande. Donc, il existe $(k-1)(5-k)$ tirages pour lesquels k est la deuxième valeur la plus petite. On a donc, pour $k \in \{2,3,4\}$

$$P(X=k) = \frac{(k-1)(5-k)}{\binom{5}{3}} = \frac{(k-1)(5-k)}{10},$$

ce qu'on peut présenter sous forme de tableau :

k	2	3	4
$P(X=k)$	3/10	4/10	3/10

Ceci est d'ailleurs conforme à ce qu'on aurait pu conjecturer *via* simulations avec le programme précédent et les instructions suivantes

```
n=5
S=zeros(1,10000)
for j=1:10000
    S(j)=X(n)
end
T=tabul(S, 'i')
bar(T(:,1), T(:,2)/10000, 'blue', width=0.1)
```



Exercice 2

(1) On simule en fait une Bernoulli. Le test est négatif si et seulement si chaque personne du pool est négative, ce qui se passe (par indépendance des contaminations) avec probabilité p^ℓ. On a donc, par exemple, le programme suivant

```
function y=pool(l,p)
    if rand()<=(1-p)^l then //si aucun n'est contaminé
        y=0;
    else //sinon, au moins un l'est et le test est positif
        y=1;
    end
endfunction
```

(2) Observons que finalement X est une binomiale de paramètre N/ℓ (le nombre de *pools*) et $1-(1-p)^\ell$ (la probabilité qu'au moins une personne du *pool* soit positive). On fait donc un test par *pool* (donc N/ℓ) et, pour chaque *pool* dont l'échantillon est positif, on refait ℓ tests, donc

$$T = \frac{N}{\ell} + \ell X.$$

```
function [X,T]=poolage_sanguin(N,p,l)
    X=grand(1,1,'bin',N/l, 1-(1-p)^l);
    T=N/l + X*l;
endfunction
```

(3) Par linéarité de l'espérance,

$$\begin{aligned} E(T) &= E\left(\frac{N}{\ell} + \ell X\right) \\ &= \frac{N}{\ell} + \ell E(X) = \frac{N}{\ell} + \ell \times \frac{N}{\ell} \times \left(1-(1-p)^\ell\right) \\ &= \frac{N}{\ell} + N\left(1-(1-p)^\ell\right) \end{aligned}$$

En particulier, la méthode du *poolage* sera avantageuse si elle permet en moyenne de faire moins de tests que la première méthode qui en fait faire N. Or

$$\begin{aligned} E(T) > N &\iff \frac{1}{\ell} + \left(1-(1-p)^\ell\right) > 1 \\ &\iff 1-p < \left(\frac{1}{\ell}\right)^{1/\ell} \end{aligned}$$